

第6讲 在 Mathematica 中作图

6-3 二元函数作图 Plot3D

Plot3D 应用对象：

在直角坐标系中绘制二元函数 $f(x, y)$ 在指定区间上的图形。

Plot3D 命令形式：

`Plot3D[f[x, y], {x, x0, x1}, {y, y0, y1}, 选项]`

在 x 和 y 的区域上，按选项画出空间曲面实数值表达式 $f(x, y)$ 的三维图形

`Plot3D[{f[x, y], g[x, y]}, {x, x0, x1}, {y, y0, y1}, 选项]`

按选项值在同一坐标系中绘制函数 $f(x, y)$ 和 $g(x, y)$ 的三维图形

1. 关于绘图区间

`Plot3D[f[x, y], {x, x0, x1}, {y, y0, y1}]` 变量 $\{x, y\}$ 在矩形区域中取值


`Plot3D[f[x, y], {x, y} ∈ reg]` 变量 $\{x, y\}$ 在几何区域 reg 取值.

例1：绘图区域是矩形区域

`Plot3D[Sin[x + y^2/3], {x, -Pi, Pi}, {y, -4.5, 4.5}]`

`Plot3D[{x^2 + y^2, 15 - x^2 - y^2}, {x, -3, 3}, {y, -3, 3}]`

`Plot3D[{x^2 + y^2, 15 - x^2 - y^2}, {x, -3, 3}, {y, -3, 3}, BoxRatios -> {1, 1, 1}]`




例2：绘图区域是单位圆。

```
Plot3D[2 x^2 + y, {x, y} ∈ Disk[]]
```

```
Plot3D[2 x^2 + y, {x, -3, 3}, {y, -5, 5}]
```

```
Plot3D[x^4 + y^4, {x, y} ∈ Disk[], ColorFunction → Function[{x, y}, Hue[x]]]
```



例3：在环形区域上绘制图形。 `RegionFunction → Function[...]`

```
Plot3D[{x^2 + y^2, 15 - x^2 - y^2}, {x, -2.2, 2.2}, {y, -2.2, 2.2},
  RegionFunction → Function[{x, y}, 0.2 < x^2 + y^2 < 4.2], BoxRatios -> {1, 1, 1}]
```

```
Plot3D[Sin[x + y^2], {x, -2, 2}, {y, -2, 2},
  RegionFunction → Function[{x, y}, 1 < x^2 + y^2 < 5]]
```

```
Plot3D[Sin[x + y^2], {x, -2, 2}, {y, -2, 2},
  RegionFunction → Function[{x, y}, x^2 + y^2 < 5]]
```

例4：用 `Exclusions` 排除矩形的部分定义域

```
Plot3D[Tan[x y] + 1 / (y^2 - x^3 + 3 x - 3), {x, -2, 2}, {y, -2, 2}]
```

```
Plot3D[Tan[x y] + 1 / (y^2 - x^3 + 3 x - 3), {x, -2, 2},
  {y, -2, 2}, Exclusions → {Cos[x y] == 0, y^2 - x^3 + 3 x - 3 == 0}]
```

```
Plot3D[Tan[x y] + 1 / (y^2 - x^3 + 3 x - 3), {x, -2, 2}, {y, -2, 2},
  BoundaryStyle → Thick, RegionFunction → Function[{x, y, z}, x^2 + y^2 >= 1]]
```

2. 关于选项


选项名	默认值	说明
Axes	True	是否绘制轴
BoundaryStyle	Automatic	如何绘制曲面的边界线
BoxRatios	{1, 1, 0.4}	有界3D box 比例
ClippingStyle	Automatic	如何绘制曲面的剪切部分
ColorFunction	Automatic	如何决定曲面的颜色
ColorFunctionScaling	True	是否用函数ColorFunction做归一化
EvaluationMonitor	None	在每次函数计算时需要计算的表达式
Exclusions	Automatic	排除的 x 、 y 曲线
ExclusionsStyle	None	如何绘制排除曲线
Filling	None	每个曲面下的填充
FillingStyle	Opacity[0.5]	填充使用的样式
MaxRecursion	Automatic	递归子划分的最大数量
Mesh	Automatic	每个方向上绘制网格线的数量
MeshFunctions	{#1 &, #2 &}	如何决定网格线的放置位置
MeshShading	None	如何设置网格线之间的阴影区域
MeshStyle	Automatic	网格线的样式
Method	Automatic	细化曲面的方式
NormalsFunction	Automatic	如何决定有效的法向量
PerformanceGoal	\$PerformanceGoal	优化执行的性能
PlotLegends	None	曲面的图例
PlotPoints	Automatic	每个方向上样本点的最初数量
PlotRange	{Full, Full, Automatic}	包括 z 范围或其它值
PlotStyle	Automatic	每个曲面样式的图形指令
PlotTheme	\$PlotTheme	overall theme for the plot
RegionFunction	(True &)	如何确定是否包含一个点
TextureCoordinateFunction	Automatic	如何确定纹理坐标
TextureCoordinateScaling	True	是否将参数缩放至范围
WorkingPrecision	MachinePrecision	内部计算使用的精度

例5 : 关于选项 BoxRatios

```
{Plot3D[x^2 - y^2, {x, -1, 1}, {y, -1, 1}],
 Plot3D[x^2 - y^2, {x, -1, 1}, {y, -1, 1}, BoxRatios -> {1, 1, 1}]}
```


例6：关于图列选项 `PlotLegends`。

```
Plot3D[{Sin[x^2 + y], Cos[x^2 + y]}, {x, 0, Pi}, {y, 0, Pi}, PlotLegends -> Automatic]
```



```
Plot3D[{Sin[x^2 + y], Cos[x^2 + y]},  
  {x, 0, Pi}, {y, 0, Pi}, PlotLegends -> {"one", "two"}]
```

```
Plot3D[{Sin[x^2 + y], Cos[x^2 + y]},  
  {x, 0, Pi}, {y, 0, Pi}, PlotLegends -> "Expressions"]
```



例7：加背景色、去网格线、去边界框。

```
Plot3D[Sin[x + y^2], {x, -3, 3}, {y, -2, 2}, Background -> Blue,  
  Mesh -> None, BoxRatios -> {1, 1, 1}, Boxed -> False]
```

例8：设置区域边界样式。

```
Plot3D[Sin[x + y^2], {x, -3, 3}, {y, -2, 2},
  RegionFunction -> Function[{x, y, z}, x^2 + y^2 >= 1],
  BoundaryStyle -> Directive[Red, Thick]] (*, Boxed -> False, Axes -> None]*)
```

例9：让曲面透明来查看内部结构。

```
Plot3D[x^2 - y^2, {x, -3, 3}, {y, -3, 3}, PlotStyle -> Opacity[0.4],
  Mesh -> None, BoxRatios -> {1, 1, 1}, Boxed -> False]
```

