第6讲 在 Mathematica 中作图

6－10 与数学相关的作图

1. NumberLinePlot 区间可视化

图示方程和不等式定义的一维区间

NumberLinePlot[{v_1, v_2, ...}] 在数轴上标出数值v_i

NumberLinePlot[*pred*, *x*] 图示数轴的区域*pred*

NumberLinePlot[*pred*, {*x*, x_{min}, x_{max}}]

在[x_{min}, x_{max}]内图示 *pred*为真的区间

例1：图示点列和区间

```
{NumberLinePlot[Prime[Range[6, 10]]], NumberLinePlot[Interval[{-1, 2}]]}
```

```
{NumberLinePlot[Prime[Range[6, 10]], Spacings → 0],
 NumberLinePlot[Interval[{0, ∞}], Spacings → 0]}
```

例2：显示不等式为真的部分

```
NumberLinePlot[x < Cos[x] / 2, {x, -0.3, 0.7}]
```

```
FindRoot[x == Cos[x] / 2, {x, 0.4}]
```

例3：图示函数的定义域

```
FunctionDomain[Sqrt[1 - x^2] + 1 / (x^2 - 1 / 4), x]
```

```
NumberLinePlot[%, x]
```

例4：显示图例的样式

```
NumberLinePlot[Table[x^2 > k, {k, 0, 5}], x, PlotTheme → {"Detailed", "Business"}]
```

```
NumberLinePlot[Table[x^2 >= k, {k, 0, 5}], x, PlotTheme → {"Detailed", "Business"}]
```

例5：图示函数增长或下降的区间

```
f[x_] := 2 x^3 + 3 x^2 - 12 x + 5
Show[{Plot[f[x], {x, -3, 3}, PlotStyle → Black], NumberLinePlot[
    {f'[x] > 0, f'[x] < 0, f'[x] == 0}, {x, -3, 3}, PlotStyle → {Red, Blue, Black},
    Spacings → 0, PlotLegends → {"增加", "减少", "稳定"}]}]
```

2. RegionPlot 和 RegionPlot3D

图示二元函数所表示的区域，三元函数所围体积

```
RegionPlot[pred, {x, xmin, xmax}, {y, ymin, ymax}]
```

```
RegionPlot3D[pred, {x, xmin, xmax}, {y, ymin, ymax}, {z, zmin, zmax}]
```

图示在绘图区域满足表达式Pred的图形.绘图区域可包含不连续部分

pred 为不等式的逻辑组合.易于实现不等式作图和隐函数作图.

例6：用不等式定义图示单位圆

```
RegionPlot[x^2 + y^2 <= 1, {x, -1, 1}, {y, -1, 1}, ColorFunction -> "SunsetColors"]
```

例7：不等式方程组作图

```
RegionPlot[x^2 + y^2 < 1 && x + y < 1, {x, -1, 1}, {y, -1, 1}, BoundaryStyle -> Dashed]
```

例8：图示两个集合的合集和交集

```
a = (x - 1/2)^2 + y^2 < 1; b = (x + 1/2)^2 + y^2 < 1;
Show[Graphics[
  {Red, PointSize[Large], Point[{0, 0}], Point[{1/2, 0}], Point[{-1/2, 0}]}],
 RegionPlot[{a, b}, {x, -1.5, 1.5}, {y, -1.5, 1.5}]]
```

```
{RegionPlot[a || b, {x, -1.5, 1.5}, {y, -1.5, 1.5}, AspectRatio → Automatic],
RegionPlot[a && b, {x, -1.5, 1.5}, {y, -1.5, 1.5}, AspectRatio → Automatic]}
```

例9：观察和比较 `x y z >= 1` 和 `x y z < 1` 的区域

```
{RegionPlot3D[x y z >= 1, {x, -5, 5}, {y, -5, 5}, {z, -5, 5}],
 RegionPlot3D[x y z < 1, {x, -5, 5}, {y, -5, 5}, {z, -5, 5}]}
```

例10：球面和锥面的交

```
RegionPlot3D[x^2 + y^2 + z^2 < 1 && x^2 + y^2 < z^2,
 {x, -1, 1}, {y, -1, 1}, {z, -1, 1}, PlotPoints → 35,
 PlotRange → All, Boxed → False, ColorFunction → "Rainbow"]
```



例11：图示牟盒方盖

```
{ta = RegionPlot3D[x^2 + y^2 ≤ 1 , {x, -1.5, 1.5}, {y, -1.5, 1.5}, {z, -1.5, 1.5}],
 tb = RegionPlot3D[ x^2 + z^2 ≤ 1, {x, -1.5, 1.5}, {y, -1.5, 1.5}, {z, -1.5, 1.5}],
 Show[ta, tb]}
```

```
RegionPlot3D[x^2 + y^2 ≤ 1 && x^2 + z^2 ≤ 1, {x, -1.5, 1.5}, {y, -1.5, 1.5},
 {z, -1.5, 1.5}, PlotPoints → 50, ColorFunction → "BlueGreenYellow", Boxed → False]
```



3. 函数可视化

DiscretePlot[*expr*, {*n*, n_{min}, n_{max}}]

图示表达式*expr*的离散值的图形，其中*n*从n_{min} 变化到n_{max}.

例12：图示3至15的斐波那契数

```
DiscretePlot[Fibonacci[k], {k, 3, 15}]
```

例13：显示黎曼和对于曲线下面积的逼近

```
T = DiscretePlot[Sin[t], {t, 0, 2 Pi, Pi / 6},
  ExtentSize → Left, PlotMarkers → "Point", AxesOrigin → {0, 0}]
```

```
Show[T, Plot[Sin[t], {t, 0, 2 Pi}]]
```



4. 向量图和流量图

VectorPlot[{vx, vy}, {x, xmin, xmax}, {y, ymin, ymax}]
图示在定义域上的二维向量图 {vx, vy}

VectorPlot3D[{vx, vy, vz}, {x, xmin, xmax}, {y, ymin, ymax}, {z, zmin, zmax}]
图示在定义域上的三维向量图 {vx, vy, vz}

StreamPlot[{v_x, v_y}, {x, xmin, xmax}, {y, ymin, ymax}]
图示向量场 {v_x, v_y} 的流线

例14：限定区域的向量图

```
{VectorPlot[{x^2, 2 y}, {x, -2, 2}, {y, -2, 2}, VectorStyle → Purple],
 VectorPlot[{x^2, 2 y}, {x, -2, 2}, {y, -2, 2},
  VectorStyle → Black, RegionFunction → Function[{x, y}, x y < 0]]}
```

例15：阻尼保守系统的全局吸引子

```
StreamPlot[{y, -y + x - x^3}, {x, -3, 3}, {y, -3, 3}, StreamScale → Large]
```

例16：两个函数的流线图及图例

```
StreamPlot[{{Sin[x], Cos[y]}, {Cos[y], -Sin[x]}},
 {x, -3, 3}, {y, -3, 3}, PlotLegends → "Expressions"]
```



5. 图论中的图

```
GraphPlot[{vi1 → vj1, vi2 → vj2, ...}]
```
生成由顶点vik到点vik的图

```
GraphPlot[m]
```
产生以邻接矩阵 m 为表示的图形

```
GraphPlot3D[{vi1 → φ1, vi2 → φ2, ...}]
```
生成三维图

例17：GraphPlot 两维图.

```
{GraphPlot[{1 -> 2, 1 -> 3, 3 -> 1, 3 -> 2, 4 -> 1, 4 -> 2}, VertexLabeling → True],
 GraphPlot[
  {1 -> 2, 1 -> 3, 3 -> 1, {3 -> 2, "3->2"}, 4 -> 1, 4 -> 2}, DirectedEdges -> True]}
```



例18：GraphPlot3D 三维图.

```
GraphPlot3D[{1 → 2, 2 → 3, 3 → 4, 1 → 4, 1 → 3, 2 → 4},
 Boxed → False, VertexLabeling → True]
```



6. 旋转曲面

```
RevolutionPlot3D[f, {t, tmin, tmax}]
```
图示绕z轴旋转曲线 z = f (t) 的旋转面，{θ, 0, 2 Pi}

```
RevolutionPlot3D[f, {t, tmin, tmax}, {θ, θmin, θmax}]
```
按θ旋转范围图示绕z轴的旋转面

例19：柱面等

```
{RevolutionPlot3D[{0.75, t}, {t, 0, 1}], RevolutionPlot3D[
  Cos[t], {t, 0, 3 Pi / 2}, Boxed → False, ColorFunction → "FallColors"]}
```

例20：单位球等

```
{RevolutionPlot3D[{Cos[t], Sin[t]}, {t, -Pi/2, Pi/2},
  ColorFunction → "Rainbow"], RevolutionPlot3D[{2 + Cos[t], Sin[t]}, {t, 0, 2 Pi}]}
```

例21：两个旋转面

```
RevolutionPlot3D[{{t, -t}, {t, t^2}}, {t, 0, 1}]  (* PlotStyle→{Yellow,Blue} *)
```




```
RevolutionPlot3D[{{t, -t}, {t, t^2}},
 {t, 0, 1}, {θ, 0, 3 Pi/2}, PlotStyle → {Pink, LightBlue}]
```