微积分和线性代数例题

例1：求极限

```
In[1]:= Limit[Fibonacci[k + 1] / Fibonacci[k], k → Infinity]
```

Out[1]= $\frac{1}{2}\left(1+\sqrt{5}\right)$

```
In[2]:= Limit[a^x / x^a, x → Infinity, Assumptions → a > 1]
```

Out[2]= ∞

```
In[3]:= Limit[a^x / x^a, x → Infinity, Assumptions → 0 < a < 1]
```

Out[3]= 0

```
In[4]:= Limit[Sin[1 / x], x -> 0]
Out[4]= Interval[{-1, 1}]
```

在7 .0 至 10.3 版本中，上例中当函数极限在 **x = 0** 不存在时，**Limit** 返回 **Interval** 对象， 表示 **sin** $(1/x)$ 在奇点 $x = 0$ 附近取值的范围. 在12 .0 的版本中输出 **Indeterminate** .

例2：求（偏）导函数 D（∂）

```
In[5]:= D[z (x^2 + y^2), {{x, y}}]
Out[5]= {2 x z, 2 y z}
```

```
In[6]:= D[z (x^2 + y^2), {x, y}]
```

Out[6]= $\partial_{\{x,y\}}\left(\left(x^2+y^2\right) z\right)$

```
In[7]:= D[z [x^2 + y^2], {{x, y}}]
```

Out[7]= $\left\{2\, x\, z'\left[x^2+y^2\right],\ 2\, y\, z'\left[x^2+y^2\right]\right\}$

```
In[8]:= D[{ x^2 y, x y^2}, {{x, y}}]
```

Out[8]= $\left\{\left\{2\, x\, y,\ x^2\right\},\ \left\{y^2,\ 2\, x\, y\right\}\right\}$

例3：求导函数 D 与全微分函数 Dt

$$\frac{\partial^2 (x^2 y)}{\partial x \partial y} \text{ 与 } \frac{d^2 (x^2 y)}{d x d y}$$

```
In[9]:= D[x^2 y, x, y]
Out[9]= 2 x
```

```
In[10]:= Dt[x^2 y, x, y]
Out[10]= 2 x + 2 y Dt[x, y] + 2 x Dt[x, y] Dt[y, x]
```

$\frac{\partial^2 (x^2 y^2)}{\partial x \partial y}$ 与 $\frac{d^2 (x^2 y^2)}{dx dy}$

In[11]:=
```
D[x^2 y^2, x, y]
```
Out[11]= 4 x y

In[12]:=
```
Dt[x^2 y^2, x, y]
```
Out[12]= 4 x y + 2 y^2 Dt[x, y] + 2 x^2 Dt[y, x] + 4 x y Dt[x, y] Dt[y, x]

例3：求积分

$$\int_0^\infty \frac{\prod_{k=0}^{8} \sin\left(\frac{x}{2k+1}\right)}{x^9} dx$$

In[13]:=
```
Integrate[Product[Sin[x / (2 k + 1)], {k, 0, 8}] / x^9, {x, 0, Infinity}]
```
Out[13]= (17 708 695 183 056 190 642 497 315 530 628 422 295 569 865 119 π) /
1 220 462 921 565 155 916 674 902 677 397 230 198 502 690 752 000 000 000

在几何区域上求积分，在这种情况下计算的是曲面积分：

In[14]:=
```
Integrate[1, {x, y, z} ∈ Sphere[]] (*  Sphere 半径为1的单位球体 *)
```
Out[14]= 4 π

In[15]:=
```
Integrate[1, {x, y, z} ∈ Cuboid[]]  (*  Cuboid 单位立方体 *)
```
Out[15]= 1

In[16]:=
```
Integrate[1, {x, y, z} ∈ Cone[]]    (*  Cone 单位圆锥体 *)
```
Out[16]= $\frac{2\pi}{3}$

例4：观察 DSolve 中 y[x] 与纯函数 y

In[17]:=
```
sol =
 DSolve[{(1 + x^2) * y''[x] + 2 x * y'[x] - 6 * x^2 - 2 == 0, y[-1] == 0, y'[-1] == 0}, y[x], x]
```
Out[17]= {{y[x] → -1 + π + x^2 + 4 ArcTan[x]}}

In[18]:=
```
sol /. x → -0.5
```
Out[18]= {{y[-0.5] → 0.537002}}

In[19]:=
```
DSolve[{(1 + x^2) * y''[x] + 2 x * y'[x] - 6 * x^2 - 2 == 0, y[-1] == 0, y'[-1] == 0}, y, x]
```
Out[19]= {{y → Function[{x}, -1 + π + x^2 + 4 ArcTan[x]]}}

得到 y 的一个纯函数解

In[20]:=

```
y''[x] + y'[x] + y[x] /. %
```

Out[20]= $\left\{1+\pi+2\,x+x^2-\frac{8\,x}{\left(1+x^2\right)^2}+\frac{4}{1+x^2}+4\,\text{ArcTan}[x]\right\}$

例5：m是一个矩阵而b是一个向量，形成增广矩阵 $\boldsymbol{m} \mid \boldsymbol{b}$：

In[21]:=
```
m = {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}};
b = {1, 2, 3};
MatrixForm[A = Transpose[Join[Transpose[m], {b}]]]
```

Out[23]//MatrixForm=

$$\begin{pmatrix} 1 & 2 & 3 & 1 \\ 4 & 5 & 6 & 2 \\ 7 & 8 & 9 & 3 \end{pmatrix}$$

再对增广矩阵做初等行变换

In[24]:=
```
MatrixForm[r = RowReduce[A]]
```

Out[24]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & -1 & -\frac{1}{3} \\ 0 & 1 & 2 & \frac{2}{3} \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

例6：三维线性空间至多有3个相互正交的向量。

In[25]:=
```
Orthogonalize[{{3, 4, 5}, {5, 4, 3}, {3, 0, 1}, {7, 8, 9}}]
```

Out[25]= $\left\{\left\{\frac{3}{5\sqrt{2}},\ \frac{2\sqrt{2}}{5},\ \frac{1}{\sqrt{2}}\right\},\ \left\{\frac{7}{5\sqrt{3}},\ \frac{1}{5\sqrt{3}},\ -\frac{1}{\sqrt{3}}\right\},\ \left\{\frac{1}{\sqrt{6}},\ -\sqrt{\frac{2}{3}},\ \frac{1}{\sqrt{6}}\right\},\ \{0,\ 0,\ 0\}\right\}$

定义连续函数的内积 $\int_{-1}^{1} f(x)\,g(x)\,dx$，将多项式空间的基函数 $\{1, x, x^2\}$

化简为一组标准正交基.

In[26]:=
```
Orthogonalize[{1, x, x^2, x^3}, Integrate[#1 #2, {x, -1, 1}] &]
```

Out[26]= $\left\{\frac{1}{\sqrt{2}},\ \sqrt{\frac{3}{2}}\,x,\ \frac{3}{2}\sqrt{\frac{5}{2}}\left(-\frac{1}{3}+x^2\right),\ \frac{5}{2}\sqrt{\frac{7}{2}}\left(-\frac{3\,x}{5}+x^3\right)\right\}$

定义连续函数的内积 $\int_{-1}^{1} x^2\, f(x)\ g(x)\ dx$，求函数f (x) 在函数g (x) 上的投影,

In[27]:=
```
Projection[f[x], g[x], Integrate[x^2 #1 #2, {x, -1, 1}] &]
```

Out[27]= $\frac{g[x]\int_{-1}^{1} x^2\, f[x]\, g[x]\, dx}{\int_{-1}^{1} x^2\, g[x]^2\, dx}$