# 第 4 讲 微积分

## 4 - 4 级数

### 4-4-1 幂级数展开

- Series[expr, {x, x0, n}]　　将expr在x = x0点展开到n阶的幂级数
- Series[expr, {x, x0, n}, {y, y0, m}]
  先对y展开到 m阶再对x展开n阶幂级数

```
Series[Sin[x^2], {x, 0, 15}]
```

```
Series[f[x], {x, a, 3}]
```

```
Series[Sin[x] * Cos[y], {x, 0, 5}, {y, 0, 5}]
```

- 展开式中可能将分数指数，对数函数等作为基本元素。

```
Series[Sqrt[x], {x, 0, 10}]
```

Series$\left[\sqrt[3]{x^2}, \{x, 0, 10\}\right]$

```
Series[Exp[Sqrt[x]], {x, 0, 6}]
```

```
Series[Exp[Sqrt[x]], {x, 1, 6}]
```

```
Series[x^x, {x, 0, 4}]
```

- Series命令可以计算无界函数在瑕点的洛朗展开式,展开式的最高次数可以是负数。

```
Series[1 / (E^x - 1), {x, 0, 10}]
```

```
Series[1 / ((E^x - 1)^10), {x, 0, -2}]
```

- Series命令可以计算函数在无穷远点的洛朗展开式

```
Series[1 / (x - 2), {x, Infinity, 4}]
```

### 4-4-2 幂级数的运算

- 幂级数求和
  - Sum[f, {i, imin, imax}]
    计算 $\sum_{i=imin}^{imax} f(i)$
  - Sum[f, {i, imin, imax, di}]
    计算 $f(imin) + f(imin + di) + f(imin + 2\,di) + \ldots f\left(imin + \left[\frac{imax}{imin}\right] di\right)$
  - Sum[f, {i, imin, imax}, {j, jmin, jmax}]　　计算 $\sum_{i=imin}^{imax}\left(\sum_{j=jmin}^{jmax} f(i, j)\right)$
- 无穷乘积

- Product[f, {i, imin, imax}]

  计算 $\prod_{i=imin}^{imax} f(i)$

```
Sum[x^i / i, {i, 1, 7}]
```

```
Sum[x^i / i, {i, 7}] (*省略求和下界时，默认从1开始 *)
```

```
Sum[x^i / i, {i, 1, 7 , 2}]
```

```
Sum[x^n / n!, {n, 0, Infinity}]
```

```
Sum[1 / i^4, {i, 1, Infinity}]
```

```
Sum[1 / (i! + (2 i)!), {i, 1, Infinity}]
```

```
N[%]
```

例：求幂级数 $\sum_{n=0}^{\infty} \frac{x^{4n+1}}{4n+1}$ 的收敛域与和函数。

```
a[n_] := 1 / (4 * n + 1);
f[x_, n_] := a[n] * x^(4 n + 1);
r = Limit[Abs[a[n + 1] / a[n]], n → Infinity]
R = 1 / r^(1 / 4)
```

检查端点是否收敛

```
Sum[f[1, n], {n, 1, Infinity}]
Sum[f[-1, n], {n, 1, Infinity}]
```

两个端点都是发散点。收敛域（-1,1）

```
Sum[f[x, n], {n, 1, Infinity}]
```

```
Sum[D[f[x, n], x], {n, 1, Infinity}]
Integrate[%, x]
```

- 给出幂级数中某一项的系数

SeriesCoefficient[f, n]　级数f中x^n的系数

SeriesCoefficient[f, {x, x0, n}]　函数f在x0的展开式中 (x - x0)^n的系数

```
SeriesCoefficient[Sin[x], {x, 0, 3}]
```

```
SeriesCoefficient[Exp[x], {x, 0, n}]
```

```
Series[Log[1 + x], {x, 0, 5}]
SeriesCoefficient[%, 2]
```

- 反函数级数

InverseSeries[s, {x, x0, n}]　给出级数s的反函数的幂级数展开式

```
t = Series[Sin[x], {x, 0, 7}]
```

```
InverseSeries[t]
```

```
Series[ArcSin[x], {x, 0, 7}]
```

- 幂级数复合

ComposeSeries[s1, s2] 表示用幂级数s2代换幂级数s1中的变量x

```
ComposeSeries[Series[Cos[x], {x, 0, 10}], Series[Sin[x], {x, 0, 10}]]
```

```
Series[Cos[Sin[x]], {x, 0, 10}]
```

```
Series[Cos[x], {x, 0, 10}] /. x → Series[Sin[x], {x, 0, 10}]
```

## 4-4-3 Fourier级数

- FourierSeries[f[x], x, n]
  计算以2Pi为周期的函数f[x]的n阶傅里叶级数 $\sum_{k=-n}^{n} c_k e^{i k t}$
  其中 $c_k = \frac{1}{2\pi} \int_{-\pi}^{\pi} f(t)\, e^{-i k t}\, dt.$
- FourierSeries[f[x, y, z ..], { x, y, z ..}, {n1, n2, n3 ..}]
  计算多元函数的傅里叶级数

```
a = FourierSeries[x^2, x, 3]
```

```
Plot[a, {x, -3 Pi, 3 Pi}]
```

```
b = FourierSeries[x * y, {x, y}, {2, 2}]
```

```
Plot3D[b, {x, -3 Pi, 3 Pi}, {y, -3 Pi, 3 Pi}]
```

对于以2L为周期的函数，要用FourierParameters->{1,2Pi/L}说明

```
FourierSeries[Abs[x], x, 3, FourierParameters → {1, 2 Pi}]
```

```
Plot[%, {x, -3, 3}]
```

- 正弦级数与余弦级数

FourierSinSeries[f[x], x, n]
f[x]是以$2\pi$为周期的函数，在$[-\pi,\pi]$上是奇函数

FourierCosSeries[f[x], x, n]
f[x]是以$2\pi$为周期的函数，在$[-\pi,\pi]$上是偶函数

若函数周期不是$2\pi$，同样用FourierParameters说明

```
c = FourierSinSeries[x, x, 5]
```

```
Plot[c, {x, -3 π, 3 π}]
```

```
d = FourierCosSeries[x^2, x, 5, FourierParameters → {1, π}]
```

```
Plot[d, {x, -4, 4}]
```