第6讲　在 Mathematica 中作图

6 - 4 三维参数、极坐标、球坐标作图

1. 三维参数函数作图

ParametricPlot3D命令形式：

ParametricPlot3D[{x (t), y (t), z (t)}, {t, t0, t1}, 选项]
在三维空间中按选项绘制空间参数曲线 {x (t), y (t), z (t)}

ParametricPlot3D[{x, y, z}, {u, u0, u1}, {v, v0, v1}, 选项]
绘制参数u和v的三维空间曲面x = x (u, v), y = y (u, v), z = z (u, v)

ParametricPlot3D[{{xa, ya, za}, {xb, yb, zb}}, {u, u0, u1}, {v, v0, v1}, 选项]
在同一坐标系中绘制两张曲面

ParametricPlot3D[{x, y, z}, {u, v} ∈ reg]　从几何区域reg取值画图

例1：画空间曲线.

```
ParametricPlot3D[{Cos[u], Sin[u], Cos[u] Sin[u]}, {u, 0, 2 Pi}]
```



例2：画二次锥面.

```
ParametricPlot3D[{r * Cos[a], r * Sin[a], r}, {a, 0, 2 Pi}, {r, -4, 4}, Boxed -> False]
```

例3：给三维参数图设置玫瑰色.

```
ParametricPlot3D[
 {2 * (Cos[o + p] + p * Sin[o + p]), 2 * (Sin[o + p] - p * Cos[o + p]), o / (2 * Pi)},
 {o, 0, 4 * Pi}, {p, 0, 4 * Pi}, PlotPoints → 40,
 BoxRatios -> {1, 1, 1.5}, Mesh -> None, ColorFunction → "RoseColors"]
```

例4：两个交叉的圆环.

```
{fx, fy, fz} = {4 + (3 + Cos[v]) Sin[u], 4 + (3 + Cos[v]) Cos[u], 4 + Sin[v]};
{gx, gy, gz} = {8 + (3 + Cos[v]) Cos[u], 3 + Sin[v], 4 + (3 + Cos[v]) Sin[u]};
ParametricPlot3D[{{fx, fy, fz}, {gx, gy, gz}}, {u, 0, 2 Pi}, {v, 0, 2 Pi},
 PlotLegends → Automatic, Boxed -> False]
```



例5：请观察参数曲线定义域的影响力.

```
{ParametricPlot3D[
  {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}, {v, 0, Pi}, {u, 0, 2 Pi}],
ParametricPlot3D[{Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}, {v, 0, Pi}, {u, 0, 16 Pi}]}
```

2. 极坐标作图

点M的极坐标由半径r和幅角θ确定，直角坐标M (x, y) 和极坐标的转换关系：

(x, y) = (r cos θ , r sinθ)。

PolarPlot 形式：

PolarPlot[r, {θ, θa, θb}]

在幅角θ定义域上绘制r = r (θ) 的曲线

例6：画出阿基米德螺线r = θ

```
{PolarPlot[θ, {θ, 0, 5 Pi}],
 PolarPlot[θ, {θ, -5 Pi, 5 Pi}], PolarPlot[θ, {θ, 0, -5 Pi}]}
```

例7：看看 $\sqrt{u}$ 在极坐标和直角坐标中的表现.

```
{PolarPlot[Sqrt[u], {u, 0, 3 Pi}, PlotStyle → {Orange, Thick}],
 Plot[Sqrt[u], {u, 0, 3 Pi}, PlotStyle → {Green, Thick}]}
```

例8：画一组极坐标曲线.

```
r = {1, 1 + 1 / 12 Sin[12 t], 1 / 2, 1 / 2 + 1 / 24 Sin[12 t]};
PolarPlot[r, {t, 0, 2 Pi}, PlotStyle → {Green, Dashed}]
```

```
PolarPlot[{Sin[5 t], Sin[4 t]}, {t, 0, 2 Pi}, PlotStyle → {Red, Purple}]
```

例9：看看函数r 在直角坐标和极坐标中的表现？

```
In[1]:= r = Exp[Cos[t - Pi / 2]] - 2 * Cos[4 * (t - Pi / 2)] + Sin[(t - Pi / 2) / 12]^5;
Plot[r, {t, 0, 36 Pi}]
```

```
In[3]:= PolarPlot[r, {t, 0, 36 Pi}, ColorFunction → "Rainbow", Axes -> None]
```

Out[3]=



3. 球坐标作图

三维空间点M的球坐标由半径r经度θ纬度φ唯一确定，其中θ 的范围从0到π，φ的范围从0 到2 π。直角坐标M (x, y, z) 和球坐标的转换关系：

(x, y, z) = (r sin θ cosφ, r sinθ sinφ, r cosθ)

```
SphericalPlot3D[r, {θ, θa, θb}, {φ, φa, φb}]
```

```
SphericalPlot3D[{r1, r2, …}, {θ, θa, θb}, {φ, φa, φb}]
```

例10：画单位球.

```
SphericalPlot3D[1, {θ, 0, Pi}, {φ, 0, 2 Pi}, Boxed -> False]
```

例11：大半球套小半球.

```
SphericalPlot3D[{1, 2, 3}, {θ, 0, Pi}, {φ, 0, 3 Pi/2},
 Boxed -> False, Axes -> False, ColorFunction → Hue[x], Axes → None]
```



例11：一束花.

```
SphericalPlot3D[Sin[7 u] Cos[7 v], {u, 0.2, Pi}, {v, 0.2, 2 Pi},
 Boxed -> False, Mesh -> None, Axes -> False, ColorFunction → "Rainbow"]
```