


第6讲 在Mathematica 中作图

6 - 1 一元函数作图Plot



Plot应用对象：在直角坐标系中绘制单变量函数 $f(x)$ 在指定区间上的图形。

Plot命令形式：

`Plot[f, {x, xmin, xmax}, 选项]`

在区间 $\{xmin, xmax\}$ 上按选项定义值绘制单变量函数 $f(x)$ 的图形

`Plot[{f1, f2, ...}, {x, xmin, xmax}, 选项]`

在区间 $\{xmin, xmax\}$ 上按选项定义同时绘制函数 $f1(x)$, $f2(x)$, ...的图形

1. 关于单变量函数 $f(x)$ ：显式函数

例：画出 $f(x) = \sin x \cos 2x$ ，在区间 $[-2\pi, 2\pi]$ 上的图形。

`Plot[Sin[x] Cos[2 x], {x, -2 Pi, 2 Pi}]`

例：在区间 $[-5, 5]$ ，在同一坐标系中画出 $\sin x$, $\cos x$, $\sin x \cos x$ 的函数曲线。

`Plot[{Sin[x], Cos[x], Sin[x] Cos[x]}, {x, -5, 5}]`

例：画出 $\left(\int \sin(x) \cos(2x) dx\right)$ ，在区间 $[-6.3, 6.3]$ 上的图形。

`Plot[Integrate[Sin[x] Cos[2 x], x], {x, -6.3, 6.3}]`

`Plot[Evaluate[Integrate[Sin[x] Cos[2 x], x]], {x, -6.3, 6.3}]`

`g = Integrate[Sin[x] Cos[2 x], x]; Plot[g, {x, -6.3, 6.3}]`

`g[x_] = Integrate[Sin[x] Cos[2 x], x]; Plot[g[x], {x, -6.3, 6.3}]`

2. 关于绘图区间

(1) 连续区间 {变量名, 表达式1, 表达式2}

例: `Plot[Sin[x], {x, -1, 1}]`

`a = 3; b = 7; Plot[Sin[y], {y, a - b, a + b}]`

(2) 画图变量在几何区域中取值 `Plot[f, x ∈ S]`

例:

`S = ImplicitRegion[x ≤ -2 || x ≥ 3, {x}]`

`Plot[Cos[x], {x} ∈ S]`

3. 关于选项: 可有可无, 可多可少.

例: 给 x 、 y 坐标轴分别加标记 " x ", " $f(x)$ ", 设置背景色为浅红色。

`Plot[(x^2 - x) Sin[x], {x, 2, 16}, AxesLabel → {"x", "f(x)"}, Background → LightRed]`

例: 给图形加上框线和网格.

`Plot[x Sin[x], {x, 0, 3}, Frame → True,
GridLines → Automatic, Background → LightBlue]`

`Plot[{Sin[x], Cos[x], Sin[x] Cos[x]}, {x, -5, 5}, PlotStyle → {Thick, Red, Dashed}]
(* , PlotLegends → "Expressions" *)`

例: 观察和比较 `Floor[x]`, `Round[x]`, `Ceiling[x]` 函数。

`Plot[Floor[x], {x, -3.5, 3.5}, Filling → Axis]`

`Plot[Round[x], {x, -3.5, 3.5}, Filling → Axis]`


`Plot[Ceiling[x], {x, -3.5, 3.5}, Filling → Axis]`

选项名	默认值	说明
AspectRatio	1 / GoldenRatio	高宽比
Axes	True	是否绘制轴
ClippingStyle	None	如何绘制曲线被剪切的区域 »
ColorFunction	Automatic	确定曲线颜色的方法
ColorFunctionScaling	True	是否缩放ColorFunction的变量
EvaluationMonitor	None	在每次函数计算时，需要计算的表达式
Exclusions	Automatic	x 中排除的点
ExclusionsStyle	None	排除点的绘制样式
Filling	None	每条曲线下填充
FillingStyle	Automatic	填充的样式
MaxRecursion	Automatic	递归子划分的最大数量
Mesh	None	在每条曲线上绘制多少个网格点
MeshFunctions	{#1 &}	如何决定网格点的放置位置
MeshShading	None	如何在网格点间绘制阴影区域
MeshStyle	Automatic	网格点的样式
Method	Automatic	修饰曲线的方法
PerformanceGoal	\$PerformanceGoal	试图优化哪些方面的性能
PlotLegends	None	曲线的图例
PlotPoints	Automatic	样本点的初始数量
PlotRange	{Full, Automatic}	y 的范围或包含的其它值
PlotRangeClipping	True	是否在曲线范围内剪切
PlotStyle	Automatic	指定每条曲线样式的图形指令
PlotTheme	\$PlotTheme	绘图的整体外观主题
RegionFunction	(True &)	如何确定是否包含一个点
TargetUnits	Automatic	在绘图中显示的单位
WorkingPrecision	MachinePrecision	内部计算使用的精度

例：填实第2条曲线到第3条曲线之间的区域。


```
Plot[{x Sin[1/x], Abs[x], -Abs[x]}, {x, -0.3, 0.3}, Filling -> {2 -> {3}}]
```

```
Plot[{x Sin[1/x], Abs[x], -Abs[x]}, {x, -0.3, 0.3},  
Filling -> {2 -> {3}}, PlotStyle -> {Thick, Dashed, Red}]
```



例：观察4种方式填实方式。

```
Table[Plot[(x^2 - x) Sin[x], {x, 2, 16}, Filling -> f], {f, {Axis, Top, Bottom, 55}}]
```



例：给曲线增彩加色。

```
Plot[Sin[x], {x, 0, 2 Pi},  
ColorFunction -> Function[{x, y}, Hue[y]], Filling -> Axis]
```

